

## Using the H8/300 to Generate a Time-of-Day Clock


### INTRODUCTION

The H8/300 series microcontrollers are highly integrated devices incorporating not only a CPU core but also a number of system peripheral devices such as on-chip ROM and RAM, a UART channel, a 16-bit Free Running Timer (FRT) module, a dual-channel 8-bit timer, several I/O Ports, and an 8-channel A/D converter

as well. Such a chip diversification certainly allows for a multitude of applications. This application note will illustrate how to implement a time-of-day digital clock using the 16-bit FRT module, utilizing I/O ports and interrupt control.

### DESCRIPTION OF OPERATION

The time-of-day clock is simulated by repeatedly pulsing 3 LEDs every second, minute, and hour, for 1/2 second. In this endeavour, the H8/330 evaluation board is utilized in order to produce the 3 time signals via port bits 0-2 of the I/O Port 5 through the corresponding output pins of jumper connector J2. Please refer to the H8/330 Evaluation Board User's Manual for details. The chip is set into operation mode 3 (single chip mode), and the controlling software is downloaded into the on-chip ROM via the RS-232 channel. An additional board is used to drive and invert the 3 time signals via a 74LS240 tri-statable inverter to the 3 pulsing LEDs, and a 1KOhm SIP resistor pack is employed to pull high the LED inputs. This configuration is illustrated in the figure below:



A 1 MHz external crystal oscillator connected across the XTAL and EXTAL pins of the controller is used, thus providing a 0.5MHz CPU clock. Bits 1 and 0 (CKS1 and 0) of the Timer Control Register (TCR) are set as to provide a 15.625KHz timer clock (0.5MHz/32).

Enabling bit 1 (OVIE) of the Timer Interrupt Enable Register (TIER) and loading the Free Running Counter (FRC) with FFFF initiates the interrupt service routine that controls the clock operation. The FRC contents are incremented on each consecutive falling edge of the timer clock. Since a compare-match value of 15,625 (or 3D09 hex) was loaded into the Output Compare Register A (OCRA), and bit 0 (CCLRA) of the Timer Control/Status Register (TCSR) was set to 1, the FRC will reset its contents to 0 and resume counting after each second has elapsed. Bit 3 (OCFA) of TCSR indicates a match between OCRA and FRC has occurred, and is continuously monitored by the user software. When it is set, a high pulse is transmitted through output port bit P5(0), and LED 0 will turn on each second. A loop counter in the program is used to hold this signal high for 1/2 second, and then turn it low. After 60 compare-matches between OCRA and FRC have occurred, output port bit P5(1) is set high, turning on LED 1 for 1/2 second, indicating that a minute has elapsed. Similarly, after 3600 compare-matches between OCRA and FRC, LED 3 will be turned on for 1/2 second via P5(3) indicating an hour has passed. Figure 1 on the next page illustrates the timing of the output port bits P5(2-0). The program flowcharts are shown in figures 2-5.

It is important to note that, in order to keep the software less complex, the timer clock has to be less than 65.536KHz (FFFF hex), which corresponds to a maximum CPU clock of  $65.536 \times 32 = 2.097\text{MHz}$ , or a maximum external oscillator frequency of 4.194MHz. This is because the OCRA and FRC are 16-bit registers and therefore cannot be loaded with a hex value greater than FFFF.




Figure 1: LED Timing




Figure 2. Main Routine.




Figure 3. Start Clock Service Routine.




Figure 4. Start Clock Service Routine (continuation)




Figure 5. Start Clock Service Routine (continuation)

```

Command line: C:\MRI\ASMH83\ASMH83.EXE -l clock.src
Line      Addr
1          ;          FRT register definitions
2
3          FF90          FRT_TIER          .equ      H'FF90
4          FF91          FRT_TCSR          .equ      H'FF91
5          FF92          FRT_FRC          .equ      H'FF92
6          FF94          FRT_OCR          .equ      H'FF94
7          FF96          FRT_TCR          .equ      H'FF96
8          FF97          FRT_TOCR          .equ      H'FF97
9
10         ;          I/O Port definitions
11
12         FFB8          P5_DDR          .equ      H'FFB8
13         FFBA          P5_DR          .equ      H'FFBA
14
15         ;          Specify interrupt vector address and program start
16
17         .org          H'24
18         .data.w       START_CLK
19         .org          H'500
20
21         ;          Perform all initializations
22
23         0500 7907 FF80          mov.w       #H'FF80,R7          ;initialize stackpointer
24         0504 0700          ldc          #0,CCR          ;unmask all interrupts
25         0506 7905 0001          mov.w       #1,R5          ;initialize decrement value for LED-on time counter
26         050A FC3C          mov.b       #60,R4L          ;set R4L with seconds counter value
27         050C F43C          mov.b       #60,R4H          ;set R4H with minutes counter value
28         050E F8FF          mov.b       #H'FF,R0L          ;configure P5 as output port
29         0510 38B8          mov.b       R0L,@P5_DDR
30         0512 7901 3D09          mov.w       #H'3D09,R1          ;load OCRA with 1 sec. count
31         0516 6B81 FF94          mov.w       #H'1E84,R2          ;set LED-on time
32         051A 7902 1E84          mov.b       #H'02,R0L          ;timer clock is 0.5MHz/32 = 15.625KHz
33         051E F802          mov.b       R0L,@FRT_TCR
34         0520 3896          mov.b       R0L,@FRT_TCSR
35         0522 7F97 7240          bclr          #4,@FRT_TOCR          ;CPU will access OCRA
36         0526 7F91 7000          bset          #0,@FRT_TCSR          ;FRC will be cleared when OCRA=FRC
37         052A 7F90 7010          bset          #1,@FRT_TIER          ;enable timer overflow interrupt request
38         052E 7900 FFFF          mov.w       #H'FFFF,R0          ;start clock
39         0532 6B80 FF92          mov.w       R0,@FRT_FRC
40         0536 0000          nop
41         0538 0000          nop
42         053A 0000          nop
43         053C 0000          nop
44
45         ;          Time-of-day clock interrupt service routine
46
47         START_CLK:
48         053E 7E91 7330          btst          #3,@FRT_TCSR          ;is OCFA set to 1?
49         0542 47FA          beq          START_CLK          ;if not, test again
50         0544 1A0C          dec          R4L          ;decrement seconds counter value
51         0546 4716          beq          MINUTES          ;if 60 sec. have passed, go to MINUTES
52         0548 7FBA 7000          bset          #0,@P5_DR          ;turn on LED 1 (indicating seconds)
53
54         LED1_ON:
55         054C 1952          sub          R5,R2          ;LED 1 will be on for 1/2 sec.
56         054E 46FC          bne          LED1_ON
57         0550 7FBA 7200          bclr          #0,@P5_DR          ;turn off LED1
58         0554 7902 1E84          mov          #H'1E84,R2          ;restore LED-on time counter value
59         0558 7F91 7230          bclr          #3,@FRT_TCSR          ;clear OCFA
60         055C 40E0          bra          START_CLK
61
62         MINUTES:
63         055E 1A04          dec          R4H          ;decrement minutes counter value
64         0560 4718          beq          HOURS          ;if 60 min. have passed, go to HOURS
65         0562 F8FB          mov.b       #H'FB,R0L          ;turn on LED 1 and 2 (indicating sec and minutes)
66         0564 38BA          mov.b       R0L,@P5_DR
67
68         LED12_ON:
69         0566 1952          sub          R5,R2          ;LED 1 and 2 will be on for 1/2 sec
70         0568 46FC          bne          LED12_ON
71         056A F8F8          mov.b       #H'F8,R0L          ;turn off LED 1 and 2
72         056C 38BA          mov.b       R0L,@P5_DR
73         056E 7902 1E84          mov.w       #H'1E84,R2          ;restore LED-on time counter value
74         0572 FC3C          mov.b       #60,R4L          ;restore seconds counter value
75         0574 7F91 7230          bclr          #3,@FRT_TCSR          ;clear OCFA
76         0578 40C4          bra          START_CLK
77
78         HOURS:
79         057A F8FF          mov.b       #H'FF,R0L          ;turn on LED 1,2, and 3
80         057C 38BA          mov.b       R0L,@P5_DR
81
82         LED123_ON:
83         057E 1952          sub          R5,R2          ;LED 1,2, and 3 will be on for 1/2 sec
84         0580 46FC          bne          LED123_ON
85         0582 F8F8          mov.b       #H'F8,R0L          ;turn off all LEDs
86         0584 38BA          mov.b       R0L,@P5_DR
87         0586 7902 1E84          mov.w       #H'1E84,R2          ;restore LED-on time counter value
88         058A F43C          mov.b       #60,R4H          ;restore minutes counter value
89         058C FC3C          mov.b       #60,R4L          ;restore seconds counter value
90         058E 7F91 7230          bclr          #3,@FRT_TCSR          ;clear OCFA
91         0592 40AA          bra          START_CLK
92         .end

```

The information contained in this document has been carefully checked, however the contents of this document may be changed and modified without notice. Hitachi America, Ltd. shall assume no responsibility for inaccuracies, or any problem involving patent infringement caused when applying the descriptions in this document. This material is protected by copyright laws. Hitachi America, Ltd. reserves all rights.